Dataplot Vol 2 Vol 1

BIWEIGHT MIDVARIANCE

Name:
BIWEIGHT MIDVARIANCE (LET)
Type:
Let Subcommand
Purpose:
Compute the biweight midvariance for a variable.
Description:
Mosteller and Tukey (see Reference section below) define two types of robustness:

1. resistance means that changing a small part, even by a large amount, of the data does not cause a large change in the estimate

2. robustness of efficiency means that the statistic has high efficiency in a variety of situations rather than in any one situation. Efficiency means that the estimate is close to optimal estimate given that we know what distribution that the data comes from. A useful measure of efficiency is:

Efficiency = (lowest variance feasible)/ (actual variance)

Many statistics have one of these properties. However, it can be difficult to find statistics that are both resistant and have robustness of efficiency.

For scale estimaors, the standard deviation (or variance) is the optimal estimator for Gaussian data. However, it is not resistant and it does not have robustness of efficiency. The median absolute deviation (MAD) is a resistant estimate, but it has only modest robustness of efficiency.

The biweight midvariance estimator is both resistant and robust of efficiency. Mosteller and Tukey recommend using the MAD or interquartile range for exploratory work where moderate efficiency in a variety of situations is adequate. The biweight midvariance estimator can be considered for situations where high performance is needed.

The biweight midvariance estimate is defined as:

$$s_{bi}^2 = \frac{\sum_{i=1}^{n}{(y - y')^2(1 - u^2)^4}} {(\sum_{i=1}^{n}{(1 - u^2)(1 - 5u^2)})^2}$$

where the summation is restricted to $$u_{i}^2 \le 1$$ and

$$y' = \mbox{median } y$$

and

$$u_{i} = \frac{y_{i} - y'}{9*MAD} \hspace{0.5in} \mbox{for } (\frac{y_{i} - y*}{cS})^{2} < 1$$

where MAD is the median absolute deviation.

Syntax:
LET <par> = BIWEIGHT MIDVARIANCE <y>
<SUBSET/EXCEPT/FOR qualification>
where <y> is the response variable;               <par> is a parameter where the computed biweight midvariance is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.
Examples:
LET A = BIWEIGHT MIDVARIANCE Y1
LET A = BIWEIGHT MIDVARIANCE Y1 SUBSET TAG > 2
Note:
A refinement of the biweight midvariance, called the biweight scale estimate, has slightly better performance than the biweight midvariance.
Note:
Dataplot statistics can be used in a number of commands. For details, enter

Default:
None
Synonyms:
None
Related Commands:
 BIWEIGHT MIDCOVARIANCE = Compute a biweight midcovariance estimate of two variable. BIWEIGHT SCALE = Compute a biweight scale estimate of a variable. BIWEIGHT LOCATION = Compute a biweight location estimate of a variable. BIWEIGHT CONFIDENCE LIMITS = Compute a biweight based confidence interval. WINSORIZED VARIANCE = Compute the Winsorized variance of a variable. MEDIAN ABSOLUTE DEVIATION = Compute the median absolute deviation of a variable. VARIANCE = Compute the variance of a variable.
Reference:
Rand Wilcox (1997), "Introduction to Robust Estimation and Hypothesis Testing," Academic Press.

Mosteller and Tukey (1977), "Data Analysis and Regression: A Second Course in Statistics," Addison-Wesley, pp. 203-209.

Applications:
Robust Data Analysis
Implementation Date:
2002/7
Program 1:
LET Y1 = NORMAL RANDOM NUMBERS FOR I = 1 1 100
LET Y2 = LOGISTIC RANDOM NUMBERS FOR I = 1 1 100
LET Y3 = CAUCHY RANDOM NUMBERS FOR I = 1 1 100
LET Y4 = DOUBLE EXPONENTIAL RANDOM NUMBERS FOR I = 1 1 100
LET A1 = BIWEIGHT MIDVARIANCE Y1
LET A2 = BIWEIGHT MIDVARIANCE Y2
LET A3 = BIWEIGHT MIDVARIANCE Y3
LET A4 = BIWEIGHT MIDVARIANCE Y4

Program 2:

MULTIPLOT 2 2
MULTIPLOT CORNER COORDINATES 0 0 100
MULTIPLOT SCALE FACTOR 2
.
LET Y1 = NORMAL RANDOM NUNBERS FOR I = 1 1 200
LET Y2 = CAUCHY RANDOM NUNBERS FOR I = 1 1 200
.
BOOTSTRAP SAMPLES 500
TITLE BIWEIGHT MIDVARIANCE BOOTSTRAP: CAUCHY
BOOTSTRAP BIWEIGHT MIDVARIANCE PLOT Y1
X1LABEL B025 = ^B025, B975=^B975
TITLE
HISTOGRAM YPLOT
.
TITLE BIWEIGHT MIDVARIANCE BOOTSTRAP: NORMAL
BOOTSTRAP BIWEIGHT MIDVARIANCE PLOT Y1
X1LABEL B025 = ^B025, B975=^B975
TITLE
HISTOGRAM YPLOT
.
END OF MULTIPLOT


NIST is an agency of the U.S. Commerce Department.

Date created: 07/26/2002
Last updated: 11/02/2015